

Using SQL in SAS

1. Perform the following tasks:

- a) Create a SQL query to display Event and Cost from pg1.storm_damage. Format the values of Cost.
- b) Add a new column named Season that extracts the year from Date.
- c) Add a WHERE clause to return rows where Cost is greater than 25 billion.
- d) Add an ORDER BY clause to arrange rows by descending Cost.
- e) Which storm had the highest cost?

```
proc sql;
select Event,
       Cost format=dollar16.,
       year(Date) as Season
from pg1.storm_damage
where Cost>25000000000
order by Cost desc;
quit;
```

Hurricane Katrina had the highest cost.

2. Perform the following tasks:

- a) Create a query to create a temporary table named top_damage.
- b) Add an additional query in the same PROC SQL step to generate a report listing all columns for the first 10 storms in the top_damage table.
- c) Add a TITLE statement before the second query to display the following text: Top 10 Storms by Damage Cost.
- d) How many of the top 10 storms occurred in 2005?

```
proc sql;
create table top_damage as
select Event,
       Date format=monyy7.,
       Cost format=dollar16.
from pg1.storm_damage
order by Cost desc;
title "Top 10 Storms by Damage Cost";
select *
from top_damage(obs=10);
quit;
```

Three of the top 10 storms occurred in 2005.

3. Perform the following tasks:

- a) Define aliases for storm_summary and storm_basincodes in the FROM clause.
- b) Use one table alias to qualify Basin in the SELECT clause.
- c) Complete the ON expression to match rows when Basin is equal in the two tables. Use the table aliases to qualify Basin in the expression. Run the code.
- d) View the report. The storm_summary table includes some lowercase Basin values. Are they in the results?

```
select Season, Name, s.Basin, BasinName, MaxWindMPH
  from pg1.storm_summary as s
    inner join pg1.storm_basincodes as b
      on s.basin=b.basin
 order by Season desc, Name;
quit;
```

The lowercase **Basin** values are not in the results because they are not matching values. To fix this, you can use the UPCASE function in the ON expression to change the values to uppercase.

```
proc sql;
select Season, Name, s.Basin, BasinName, MaxWindMPH
  from pg1.storm_summary as s
    inner join pg1.storm_basincodes as b
      on upcase(s.basin)=b.basin
 order by Season desc, Name;
quit;
```